

Description

A range plot has two y variables, such as high and low daily stock prices or upper and lower 95% confidence limits.

`twoway rbar` plots a range, using bars to connect the high and low values.

Also see [\[G-2\] graph bar](#) for more traditional bar charts.

Quick start

Range plot with bars

```
twoway rbar y1 y2 x
```

Same as above, but with horizontal bars

```
twoway rbar y1 y2 x, horizontal
```

Specify bars to be 0.75 times the default width to create space between bars

```
twoway rbar y1 y2 x, barwidth(.75)
```

Overlay a line plot of y_3 versus x

```
twoway rbar y1 y2 x || line y3 x
```

Same as above, but with light gray bars

```
twoway rbar y1 y2 x, color(gs10) || line y3 x
```

Add the title “My Title” to the graph

```
twoway rbar y1 y2 x, color(gs10) || line y3 x, title("My Title")
```

Menu

Graphics > Two-way graph (scatter, line, etc.)

Syntax

```
twoway rbar y1var y2var xvar [if] [in] [, options]
```

<i>options</i>	Description
<code>vertical</code>	vertical bars; the default
<code>horizontal</code>	horizontal bars
<code>barwidth(#)</code>	width of bar in <i>xvar</i> units
<code>mwidth</code>	use <code>msize()</code> rather than <code>barwidth()</code>
<code>msize(<i>markersizestyle</i>)</code>	width of bar
<code>barlook_options</code>	change look of bars
<code>colorvar_options</code>	change color of bars based on values of a variable
<code>axis_choice_options</code>	associate plot with alternative axis
<code>twoway_options</code>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

Options `barwidth()`, `mwidth`, and `msize()` are *rightmost*, and vertical and horizontal are *unique*; see [G-4] **Concept: repeated options**.

Options

`vertical` and `horizontal` specify whether the high and low *y* values are to be presented vertically (the default) or horizontally.

In the default `vertical` case, *y1var* and *y2var* record the minimum and maximum (or maximum and minimum) *y* values to be graphed against each *xvar* value.

If `horizontal` is specified, the values recorded in *y1var* and *y2var* are plotted in the *x* direction and *xvar* is treated as the *y* value.

`barwidth(#)` specifies the width of the bar in *xvar* units. The default is `barwidth(1)`. When a bar is plotted, it is centered at *x*, so half the width extends below *x* and half above.

`mwidth` and `msize(markersizestyle)` change how the width of the bars is specified. Usually, the width of the bars is determined by the `barwidth()` option documented below. If `mwidth` is specified, `barwidth()` becomes irrelevant and the bar width switches to being determined by `msize()`. This all has to do with the units in which the width of the bar is specified.

By default, bar widths are specified in the units of *xvar*, and if the `barwidth()` option is not specified, the default width is 1 *xvar* unit.

`mwidth` specifies that you wish bar widths to be measured in size units; see [G-4] **size**. When you specify `mwidth`, the default changes from being 1 *xvar* unit to the default width of a marker symbol.

If you also specify `msize()`, the width of the bar is modified to be the size specified.

`barlook_options` set the look of the bars. The most important of these options is `color(colorstyle)`, which specifies the color and opacity of the bars; see [G-4] **colorstyle** for a list of color choices. See [G-3] **barlook_options** for information on the other `barlook_options`.

`colorvar_options` specify that the color of the bars be determined by the levels of the numeric variable *colorvar*; see [G-3] **colorvar_options**.

`axis_choice_options` associate the plot with a particular y or x axis on the graph; see [G-3] `axis_choice_options`.

`twoway_options` are a set of common options supported by all `twoway` graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over by() groups, and change some advanced settings. See [G-3] `twoway_options`.

Remarks and examples

Remarks are presented under the following headings:

Typical use
Advanced use

Typical use


We have daily data recording the values for the S&P 500 in 2001:

```
. use https://www.stata-press.com/data/r19/sp500
(S&P 500)
. list date high low close in 1/5
```

	date	high	low	close
1.	02jan2001	1320.28	1276.05	1283.27
2.	03jan2001	1347.76	1274.62	1347.56
3.	04jan2001	1350.24	1329.14	1333.34
4.	05jan2001	1334.77	1294.95	1298.35
5.	08jan2001	1298.35	1276.29	1295.86

We will use the first 57 observations from these data:

```
. twoway rbar high low date in 1/57, barwidth(.6)
```



We specified `barwidth(.6)` to reduce the width of the bars. By default, bars are 1 x unit wide (meaning 1 day in our data). That default resulted in the bars touching. `barwidth(.6)` reduced the width of the bars to 0.6 days.

Advanced use

The useful thing about `twoway rbar` is that it can be combined with other twoway plottypes (see [G-2] [graph twoway](#)):

```
. twoway rbar high low date, barwidth(.6) color(gs7) ||  
  line close date || in 1/57
```



There are two things to note in the example above: our specification of `color(gs7)` and that we specified that the range bars be drawn first, followed by the line. We specified `color(gs7)` to tone down the bars: By default, the bars were too bright, making the line plot of close versus date all but invisible. Concerning the ordering, we typed

```
. twoway rbar high low date, barwidth(.6) color(gs7) ||  
  line close date || in 1/57
```

so that the bars would be drawn first and then the line drawn over them. Had we specified

```
. twoway line close date ||  
  rbar high low date, barwidth(.6) color(gs7) || in 1/57
```

the bars would have been placed on top of the line and thus would have occulted the line.

References

- Cox, N. J. 2016. [Speaking Stata: Multiple bar charts in table form](#). *Stata Journal* 16: 491–510.
- Kohler, U., and C. Brzinsky-Fay. 2005. [Stata tip 25: Sequence index plots](#). *Stata Journal* 5: 601–602.

Also see

- [G-2] **graph twoway bar** — Two-way bar plots
- [G-2] **graph twoway rarea** — Range plot with area shading
- [G-2] **graph twoway rcap** — Range plot with capped spikes
- [G-2] **graph twoway rcapsym** — Range plot with spikes capped with marker symbols
- [G-2] **graph twoway rconnected** — Range plot with connected lines
- [G-2] **graph twoway rline** — Range plot with lines
- [G-2] **graph twoway rscatter** — Range plot with markers
- [G-2] **graph twoway rspike** — Range plot with spikes

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

